Using Functions SYSFUNC and IFC to Conditionally Execute Statements in Open Code
Ronald J. Fehd, Centers for Disease Control and Prevention, Atlanta, GA, USA

ABSTRACT
This paper explains how to conditionally execute statements in open code without having to wrap the code in a macro. This is accomplished by combining the macro function `sysfunc` with the data step function `ifc`. The result is that parameterized include programs gain the power of conditional processing — `%if ... %then` — for which macros are usually used. This is another in the Journeymen's Tools series.

Audience advanced users and macro programmers.

Keywords assertions, function, ifc, parameterized include program, sysfunc

In this paper
This paper contains the following topics.

<table>
<thead>
<tr>
<th>Section</th>
<th>Page</th>
</tr>
</thead>
<tbody>
<tr>
<td>Introduction</td>
<td>2</td>
</tr>
<tr>
<td>Function IFC</td>
<td>3</td>
</tr>
<tr>
<td>Function SysFunc</td>
<td>4</td>
</tr>
<tr>
<td>SysFunc and IFC</td>
<td>4</td>
</tr>
<tr>
<td>Usage Examples</td>
<td>5</td>
</tr>
<tr>
<td>Calling Routines</td>
<td>6</td>
</tr>
<tr>
<td>Assertions</td>
<td>8</td>
</tr>
<tr>
<td>Conclusion</td>
<td>10</td>
</tr>
</tbody>
</table>
INTRODUCTION

Overview

Program development proceeds through these stages:

- create ad hoc programs
- identify programs with similar algorithms
- consolidate into parameterized include program
- where necessary, convert to macro for conditional processing

The techniques shown here eliminate the need for macros and replace conditional processing with inline statements.

Example: %if statement used to conditionally execute blocks of code.

```sas
%macro DoThis;
%if <condition> %then %do;
  * statements for true;
  %end;
%else %do;
  * statements for false;
  %end;
run;%mend;
%DoThis
```

can be replaced with this:

```sas
%sysfunc(ifc(<condition>
    ,* statements for true;
    ,* statements for false;
  ))
```
The `ifc` function has four parameters:

1. a logical expression
2. character value returned when true
3. value returned when false
4. value returned when missing, which is optional

```sas
IFC( logical-expression , value-returned-when-true , value-returned-when-false <, value-returned-when-missing>
```

This example shows the function in a data step.

```sas
DATA TestIfc;
 attrib N length = 4
 Value length = $ %length(missing);
 do N = ., 0, 1;
   Value = ifc(N , 'true' , 'false' , 'missing' );
   putlog N= Value=;
 output;
 end;
 stop;
 run;
 Proc Print data = TestIfc;
```

```csv
Obs N Value
--- - -------
 1 .  missing
 2 0  false
 3 1  true
```
**FUNCTION SYSFUNC**

The macro function sysfunc provides a method to access certain data step functions and the values of options.

**SysFunc and GetOption**

This example shows the combination of sysfunc and getoption writing a note to the log with the value of the option source2.

```plaintext
put-sysfunc-getoption.log
1 options nosource2;
2 %Put option source2 is %sysfunc(getoption(source2));
3 option source2 is NOSOURCE2
4 options source2;
5 %Put option source2 is %sysfunc(getoption(source2));
6 option source2 is SOURCE2
```

**SYSFUNC AND IFC**

This example contains the previous usage of sysfunc and getoption. Its purpose is to show that the evaluation of the logical expression can produce conditional text.

```plaintext
put-sysfunc-ifc.log
1 options nosource2;
2 %Put option source2 is
3 %sysfunc(ifc(%sysfunc(getoption(source2)) eq SOURCE2
4 ,True
5 ,False
6 ,Missing
7 ) );
8 option source2 is False
```

Note that the sysfunc(ifc(...)) is within the %put statement whose ending semicolon is on log line 7.

```plaintext
sysfunc-ifc-error.log
1 options nosource2;
2 %sysfunc(ifc(%sysfunc(getoption(source2)) eq SOURCE2
3 ,%Put True;
4 True
5 ,%Put False;
6 False
7 ,%Put Missing;
8 Missing
9 ))
```

What happened here? All the macro %put statements were executed!
This problem is the same as with call execute; which is explained by Fehd and Carpenter [4], sgf2007.113. The statements must be wrapped in macro function nrstr (No Rescan String) which delays their execution until after the ifc condition has been evaluated.

```
sysfunc-ifc-nrstr.log
1 1 options nosource2;
2 2 %sysfunc(ifc(%sysfunc(getoption(source2)) eq SOURCE2 
3 3 ,%nrstr(%Put True;)) 
4 4 ,%nrstr(%Put False;)
5 5 ,%nrstr(%Put Missing;)
6 6 ) ) 
7 False
```

Note that the sysfunc(ifc(. . .)) is not a statement; the function calls are completed by the two closing parentheses on log line 6.

### USAGE EXAMPLES

#### Overview
These are common tests:
- boolean: zero, one or greater than one
- comparison of two values
- existence

#### Boolean
Many functions return either the set of boolean values — (0,1) — or zero and a negative or positive integer which evaluates as true.

```
0,1 %sysfunc(cexist(&Catalog.))
0,1, ge 1 &Nobs.
-1,0,1, ge 1 %sysfunc(attrn(&dsid.,nobs))
```

#### Comparison
Any logical expression can be used for the first argument of the ifc function.

```
&NobsData1. eq &NobsData2.
&Nobs. gt 100
%sysfunc(fileref(&FileRef.)) eq 0
%upcase(%substr(&Type.,1,1)) eq C or &Type. eq 2
```

There are pitfalls when comparing user-supplied values; refer to Fehd [2], sas-wiki.Cond-Exec-Global-Stmnts for examples and fixes.
Test Suite

Each of the following programs has been tested as a parameterized include file, illustrated by this example:

```sas
options source2;
Proc Format library = Work; value Test 0='zero';run;
%Let Catalog = Work.Formats;
%Include Project(assert-cexist-catalog);
%Let Catalog = Library.Formats;
%Include Project(assert-cexist-catalog);
```

Both the program `*.sas` and its test `*-Test.sas` are available in a .zip at Fehd [2], sas-wiki.Cond-Exec-Global-Stmtns

CALLING ROUTINES

Branching with Includes

Evaluation of the logical expression can be used to branch to the execution of either of two programs.

```sas
* cond-inc-which;
options source2;
%Let Data = sashelp.Class;
%*Let Data = sashelp.ClassX;/* test not exist;
%let dsid = %sysfunc(open(&Data.));
%let Nobs = %sysfunc(ifc(&DsId.,%nrstr( %sysfunc(attrn(&dsid.,nobs)); ) ,%nrstr(0) ));
%sysfunc(ifc(&Nobs.,%nrstr(%Include Project(cond-inc-1);) ,%nrstr(%Include Project(cond-inc-0);) ))
```

```
cond-inc-1.sas
%Put Note2: file is cond-inc-1;
```

```
cond-inc-0.sas
%Put Note2: file is cond-inc-0;
```

Procedure By Type

Evaluation of the logical expression can be used to branch to the execution of either of two sets of statements.

The elaborate test of the value of Type ensures that this code can be called with a list processing data set created by either Proc Contents whose Type are in (1,2) and Proc SQL whose Type are in (‘char’,’num’).

```sas
title3 "proc-type of &Libname..&Memname.;";
%sysfunc(ifc( %upcase(%substr(&Type.,1,1)) eq C
   or &Type eq 2
   ,%nrstr(Proc Freq data = &Libname..&Memname.;
             tables &Name.;)
   ,%nrstr(Proc Means data = &Libname..&Memname.;
             var &Name.;) ))
run;
```

```
proc-type.sas
```
This code is similar to the conditional processing in Fehd [1], sugi25.038, the SmryEachVar data review suite.

```sas
options source2;
%Let Libname = sashelp;
%Let Memname = Class;
%Let Name = Sex;
%Let Type = C;
%Include Project(proc-type);

%Let Name = Height;
%Let Type = 1;
%Include Project(proc-type);
```

```sas
options source2;
%Let Libname = sashelp;
%Let Memname = Class;
Proc SQL;
select catt('%let Name =',Name,';' ,'
    ', '%let Type =',Type,';' ,'
    ', '%Include Project(proc-type)')
into :List separated by ';' from Dictionary.Columns
where Libname eq "%upcase(&Libname.)"
    and Memname eq "%upcase(&Memname.)"
    and Memtype eq 'DATA';
quit;
&List.;
```

```sas
options source2;
%Let Libname = sashelp;
%Let Memname = Class;
Proc Contents data = &Libname..&Memname.
    noprint
    out = Work.Contents;
DATA _Null_;
attrib Stmnt length = $72;
do until(EndoFile);
    set Work.Contents end = EndoFile;
    Stmnt = catt('%Let Name =',Name,';');
    link ExecStmt;
    Stmnt = catt('%Let Type =',Type,';');
    link ExecStmt;
    Stmnt = '%Include Project(proc-type)';
    link ExecStmt;
end;
ExecStmt: call execute(catt('%nrstr(',Stmnt,')'));
return;
stop;
run;
```
Number of Rows

The `attrn` function can be used to get the number of observations of a data set. The data set must be opened before the function can return the value. If the data set exists then the function can be applied; otherwise when the open of the data set fails, it is not necessary to close the data set.

The values returned by `attrn(...,nobs)` are in:

-1 for a view

0 empty

ge 1 number of rows

```sas
%let dsid = %sysfunc(open(&Data. ));
%let Nobs =
%sysfunc(ifc(&DsId.
  ,%nrstr( %sysfunc(attrn(&dsid.,nobs));
  ,%let rc = %sysfunc(close(&dsid.)); })
  ,%nrstr(0) ));
%sysfunc(ifc(&Nobs.
  ,%nrstr(%Put Note2: Data &Data. available Nobs: &Nobs.;)
  ,%nrstr(%Put %sysfunc(sysmsg());
  endSAS;))}
```

This may seem strange, but minus one evaluates as true.

```sas
DATA Work.ClassView / view = Work.ClassView;
set sashelp.class(obs=1);
run;
NOTE: DATA STEP view saved on file WORK.CLASSVIEW.

%Let Data = Work.ClassView;
%Include Project(assert-nobs);
Note2: Data Work.ClassView available Nobs: -1
```

**ASSERTIONS**

**Concept**

Assertions are tests at the beginning of a program that determine whether the program input parameters are valid. The assertion of existence follows the rules of simple boolean evaluation: the functions return zero or one.

**Example of Assertion Usage**

* name: MyProgram;
  * confirm input exists;
%let Object = ObjectName;
%Include Asserts(ExistObject);* false==endSAS;
*processing;

**Exist Catalog**

The `cexist` function can be used to test the existence of a catalog.

```sas
%sysfunc(ifc(%sysfunc(cexist(&Catalog.)))
  ,%nrstr(%Put Note2: Catalog &Catalog. exists;)
  ,%nrstr(%Put Note3: not exist &Catalog.;
  endSAS;))
```
Exist Data

The `exist` function can be used to test whether a data set exists.

```sas
%sysfunc(ifc(%sysfunc(exist(&Data.)))
  ,%nrstr(%Put Note2: Data &Data. exists;)
  ,%nrstr(%Put Note3: not exist &Data.;
    endSAS; ) )
```

Exist Fileref

The `fexist` function can be used to test the existence of a fileref which is a directory-specification or folder.

```sas
%sysfunc(ifc(%sysfunc(fexist(&FileRef.)))
  ,%nrstr(%Put Note2: FileRef &FileRef. exists;)
  ,%nrstr(%Put Note3: not exist &FileRef.;
    endSAS; ) )
```

Exist File

The `fileexist` function can be used to test the existence of a file-specification.

```sas
%sysfunc(ifc(%sysfunc(fileexist(&FileNameExt.)))
  ,%nrstr(%Put Note2: FileNameExt &FileNameExt. exists;)
  ,%nrstr(%Put Note3: not exist &FileNameExt.;
    endSAS; ) )
```

Exist Fileref and File

The `fileref` function is a special case. Instead of returning one as success and zero as failure, it returns this set of values:

-1 fileref exists but file associated with the fileref does not exist
0 fileref and external file both exist
+1 fileref not assigned

Thus success is defined as `eq 0`

```sas
%sysfunc(ifc(%sysfunc(fileref(&FileRef.)) eq 0
  ,%nrstr(%Put Note2: FileRef &FileRef. exists;)
  ,%nrstr(%Put Note3: not exist &FileRef.;
    endSAS; ) )
```

For an example of usage of function `fileref` see the OnLine Doc for function `finfo`.

Summary of Assertion Usage

Each of these examples show the failure as sufficient to terminate — `endSAS` — the program. In practice consider adding other programs which provide information written to a log or data set. Nelson et al. [6], sesug2003.001 discuss a system of event management where subroutines log program completion and e-mail notifications of failure are sent to interested users.
Conclusion

The combination of functions `sysfunc` and `ifc` can be used to reduce the need for macros where only conditional processing is needed.

Further Reading

**Assertions** Nelson et al. [6], `sesug2003.001` describe assertions in an event management and notification system.

**Call Execute and Nrstr** Fehd and Carpenter [4], `sgf2007.113` demonstrate the problem of using call execute to generate macro calls and how using function `nrstr` solves that problem.

**Conditional Executions of Includes** Fehd [1], `sugi25.038` provides a trick for conditional execution of included programs.

**Macro IFF** Henderson [5], `saspress.60282` provides a macro to get around the problem of user-supplied values in comparisons.

**Using Parameterized Includes** Fehd [1], `sugi25.038` provides a data review suite of 80 programs with modules, routines and subroutines of parameterized includes.

REFERENCES


The code examples in this paper are available on the web at Fehd [2], `Sas-wiki.Cond-Exec-Global-Stmnts` SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration.